

# NAG Library Routine Document

## C09AAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

C09AAF returns the details of the chosen one-dimensional discrete wavelet filter. For a chosen mother wavelet, discrete wavelet transform type (single-level or multi-level DWT or MODWT) and end extension method, this routine returns the maximum number of levels of resolution (appropriate to a multi-level transform), the filter length, and the number of approximation coefficients (equal to the number of detail coefficients) for a single-level DWT or MODWT or the total number of coefficients for a multi-level DWT or MODWT. This routine must be called before any of the one-dimensional discrete transform routines in this chapter.

### 2 Specification

```
SUBROUTINE C09AAF (WAVNAM, WTRANS, MODE, N, NWLMAX, NF, NWC, ICOMM,      &
                  IFAIL)
INTEGER          N, NWLMAX, NF, NWC, ICOMM(100), IFAIL
CHARACTER(*)    WAVNAM
CHARACTER(1)    WTRANS, MODE
```

### 3 Description

One-dimensional discrete wavelet transforms (DWT) or maximum overlap wavelet transforms (MODWT) are characterised by the mother wavelet, the end extension method and whether multiresolution analysis is to be performed. For the selected combination of choices for these three characteristics, and for a given length,  $n$ , of the input data array,  $x$ , C09AAF returns the dimension details for the transform determined by this combination. The dimension details are:  $l_{\max}$ , the maximum number of levels of resolution that that could be computed were a multi-level DWT/MODWT applied;  $n_f$ , the filter length;  $n_c$  the number of approximation (or detail) coefficients for a single-level DWT/MODWT or the total number of coefficients generated by a multi-level DWT/MODWT over  $l_{\max}$  levels. These values are also stored in the communication array ICOMM, as are the input choices, so that they may be conveniently communicated to the one-dimensional transform routines in this chapter.

### 4 References

None.

### 5 Parameters

- 1: WAVNAM – CHARACTER(\*) *Input*  
*On entry:* the name of the mother wavelet. See the C09 Chapter Introduction for details.  
WAVNAM = 'HAAR'  
Haar wavelet.  
WAVNAM = 'DB $n$ ', where  $n = 2, 3, \dots, 10$   
Daubechies wavelet with  $n$  vanishing moments ( $2n$  coefficients). For example,  
WAVNAM = 'DB4' is the name for the Daubechies wavelet with 4 vanishing moments  
(8 coefficients).

WAVNAM = 'BIOR $x.y$ ', where  $x.y$  can be one of 1.1, 1.3, 1.5, 2.2, 2.4, 2.6, 2.8, 3.1, 3.3, 3.5 or 3.7

Biorthogonal wavelet of order  $x.y$ . For example WAVNAM = 'BIOR3.1' is the name for the biorthogonal wavelet of order 3.1.

*Constraint:* WAVNAM = 'HAAR', 'DB2', 'DB3', 'DB4', 'DB5', 'DB6', 'DB7', 'DB8', 'DB9', 'DB10', 'BIOR1.1', 'BIOR1.3', 'BIOR1.5', 'BIOR2.2', 'BIOR2.4', 'BIOR2.6', 'BIOR2.8', 'BIOR3.1', 'BIOR3.3', 'BIOR3.5' or 'BIOR3.7'.

2: WTRANS – CHARACTER(1) *Input*

*On entry:* the type of discrete wavelet transform that is to be applied.

WTRANS = 'S'

Single-level decomposition or reconstruction by discrete wavelet transform.

WTRANS = 'M'

Multiresolution, by a multi-level DWT or its inverse.

WTRANS = 'T'

Single-level decomposition or reconstruction by maximal overlap discrete wavelet transform.

WTRANS = 'U'

Multi-level resolution by a maximal overlap discrete wavelet transform or its inverse.

*Constraint:* WTRANS = 'S', 'M', 'T' or 'U'.

3: MODE – CHARACTER(1) *Input*

*On entry:* the end extension method. Note that only periodic end extension is currently available for the MODWT.

MODE = 'P'

Periodic end extension.

MODE = 'H'

Half-point symmetric end extension.

MODE = 'W'

Whole-point symmetric end extension.

MODE = 'Z'

Zero end extension.

*Constraints:*

MODE = 'P', 'H', 'W' or 'Z' for DWT;

MODE = 'P' for MODWT.

4: N – INTEGER *Input*

*On entry:* the number of elements,  $n$ , in the input data array,  $x$ .

*Constraint:*  $N \geq 2$ .

5: NWLMAX – INTEGER *Output*

*On exit:* the maximum number of levels of resolution,  $l_{\max}$ , that can be computed when a multi-level discrete wavelet transform is applied. It is such that  $2^{l_{\max}} \leq n < 2^{l_{\max}+1}$ , for  $l_{\max}$  an integer.

6: NF – INTEGER *Output*

*On exit:* the filter length,  $n_f$ , for the supplied mother wavelet. This is used to determine the number of coefficients to be generated by the chosen transform.

7: NWC – INTEGER

*Output*

*On exit:* for a single-level transform (WTRANS = 'S' or 'T'), the number of approximation coefficients that would be generated for the given problem size, mother wavelet, extension method and type of transform; this is also the corresponding number of detail coefficients. For a multi-level transform (WTRANS = 'M' or 'U') the total number of coefficients that would be generated over  $l_{\max}$  levels and with KEEPA = 'A' for MODWT.

8: ICOMM(100) – INTEGER array

*Communication Array*

*On exit:* contains details of the wavelet transform and the problem dimension which is to be communicated to the one-dimensional discrete discrete transform routines in this chapter.

9: IFAIL – INTEGER

*Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, WAVNAM =  $\langle value \rangle$  was an illegal value.

IFAIL = 2

On entry, WTRANS =  $\langle value \rangle$  was an illegal value.

IFAIL = 3

On entry, MODE =  $\langle value \rangle$  was an illegal value.

On entry, WTRANS = 'T' or 'U' and MODE  $\neq$  'P'.

Constraint: MODE = 'P' when WTRANS = 'T' or 'U'.

IFAIL = 4

On entry, N =  $\langle value \rangle$ .

Constraint:  $N \geq 2$ .

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

Not applicable.

## 9 Further Comments

None.

## 10 Example

This example computes the one-dimensional multi-level resolution for 8 values by a discrete wavelet transform using the Haar wavelet with zero end extensions. The length of the wavelet filter, the number of levels of resolution, the number of approximation coefficients at each level and the total number of wavelet coefficients are printed.

### 10.1 Program Text

```

Program c09aafe

!      C09AAF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: c09aaf, c09ccf, c09cdf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                    :: ifail, lenc, n, nf, nnz, nwc,      &
                          nwlmax, ny
Character (10)             :: mode, wavnam, wtrans
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: c(:), x(:), y(:)
Integer, Allocatable       :: dwtlev(:)
Integer                    :: icomm(100)
!      .. Intrinsic Procedures ..
Intrinsic                  :: sum
!      .. Executable Statements ..
Write (nout,*) 'C09AAF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
!      Read problem parameters
Read (nin,*) n
Read (nin,*) wavnam, mode
Allocate (x(n),y(n))

Write (nout,99999) wavnam, mode, n

!      Read data array and write it out

Read (nin,*) x(1:n)

```

```

Write (nout,*) ' Input Data           X : '
Write (nout,99998) x(1:n)

! Query wavelet filter dimensions
! For Multi-Resolution Analysis, decomposition, wtrans = 'M'
wtrans = 'Multilevel'

! ifail: behaviour on error exit
!       =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call c09aaf(wavnam,wtrans,mode,n,nwlmax,nf,nwc,icomm,ifail)

lenc = nwc
Allocate (c(lenc),dwtlev(nwlmax+1))

! Perform Discrete Wavelet transform
ifail = 0
Call c09ccf(n,x,lenc,c,nwlmax,dwtlev,icomm,ifail)

Write (nout,*)
Write (nout,99997) nf
Write (nout,99996) nwlmax
Write (nout,99995)
Write (nout,99994) dwtlev(1:nwlmax+1)
Write (nout,99993) nwc
nnz = sum(dwtlev(1:nwlmax+1))
Write (nout,*)
Write (nout,99992)
Write (nout,99998) c(1:nnz)

! Reconstruct original data
ny = n

ifail = 0
Call c09cdf(nwlmax,lenc,c,ny,y,icomm,ifail)

Write (nout,*)
Write (nout,99991)
Write (nout,99998) y(1:ny)

99999 Format (1X,' Parameters read from file :: '/'      Wavelet : ',A10, &
' End mode : ',A10,' N = ',I10)
99998 Format (8(F8.3,1X):)
99997 Format (1X,' Length of wavelet filter :           ',I10)
99996 Format (1X,' Number of Levels :                   ',I10)
99995 Format (1X,' Number of coefficients in each level : ')
99994 Format (16X,8(I8,1X):)
99993 Format (1X,' Total number of wavelet coefficients : ',I10)
99992 Format (1X,' Wavelet coefficients C : ')
99991 Format (1X,' Reconstruction           Y : ')
End Program c09aaf

```

## 10.2 Program Data

```

C09AAF Example Program Data
8           : n
Haar Zero  : wavnam, mode
2.0
5.0
8.0
9.0
7.0
4.0
-1.0
1.0       : x(1:n)

```

### 10.3 Program Results

C09AAF Example Program Results

```

Parameters read from file ::
  Wavelet : Haar      End mode : Zero      N =          8
Input Data      X :
  2.000    5.000    8.000    9.000    7.000    4.000    -1.000    1.000

Length of wavelet filter :          2
Number of Levels :          3
Number of coefficients in each level :
          1          1          2          4
Total number of wavelet coefficients :      8

Wavelet coefficients C :
12.374    4.596    -5.000    5.500    -2.121    -0.707    2.121    -1.414

Reconstruction      Y :
  2.000    5.000    8.000    9.000    7.000    4.000    -1.000    1.000

```

---